TYMSHARE TYMCOM-X MANUALS
REFERENCE SERIES

SPICE

OCTOBER 1972

TYMSHARE, INC.
10340 Bubb Road
Cupertino, California 95014

TYMSHARE®

This document was written by L. W. Nagel and D. O. Pederson of the Department of Electrical Engineering, College of Engineering, University of California.

CONTENTS

Page

Program Use 1

Program Limitations 1

Types of Analyses 2
- DC Analysis 2
- AC Small Signal Analysis 2
- Transient Analysis 3
- Analysis at Different Temperatures 3

Convergence 4

Input Format 4

Circuit Description 5

Circuit Statements 5
- Resistors, Capacitors, Inductors 5
- Voltage Controlled Current Sources 5
- Independent Sources 6
  - PULSE 7
  - EXPONENTIAL 7
  - SINUSOIDAL 8
- Bipolar Junction Transistors 8
- Junction Diodes 9
- Junction Field Effect Transistors 9
- MOSFETs 9
- .MODEL Statement 9
  - Diode Models 10
  - Ebers-Moll BJT Models 11
  - Gummel-Poon BJT Models 11
  - JFET Models 12
  - MOSFET Models 13

Control Statements 14
- Title Statement 14
- .END Statement 14
- Comment Statement 14
- No Print Statement 14
- .TEMP Statement 15
- .OUTPUT Statement 15
- .DC Statement 16
- .AC Statement 17
- .TRAN Statement 17

Sample Data Files 18

Sample Problem 20

SPICE is a general purpose circuit simulation program for nonlinear DC, nonlinear transient, and linear AC analysis. Circuits may contain resistors, capacitors, inductors, independent voltage and current sources, voltage dependent current sources, and the four most common semiconductor devices: BJTs, diodes, JFETs, and MOSFETs.

SPICE has built-in models for the semiconductor devices, and the user specifies only the pertinent model parameter values. Two models are available for the BJT. The simpler model is based on the Ebers-Moll model and includes charge storage effects, ohmic resistances, and a current dependent output conductance. A model based on the integral charge model of Gummel and Poon is also available for problems which require a more sophisticated BJT model. The diode model can be used for either junction diodes or Shottky barrier diodes. The JFET and MOSFET models are both based on the FET model of Hodges and Schichman.

PROGRAM USE

SPICE is extremely easy to use. The user simply creates a data file in EDITOR containing circuit statements and control statements. (See the Tymshare EDITOR Reference Manual for a complete description of EDITOR.)

The user then calls SPICE by typing R SPICE and a Carriage Return in TYMEX. The program requests the name of the input file and then prints the results of the analysis. A sample program is included at the end of this document.

PROGRAM LIMITATIONS

400 Nodes, including integral device nodes. Each nonzero ohmic resistance in a device will generate an internal node. For example, a circuit with 35 user specified nodes and 10 BJTs with nonzero base and collector resistances will contain 55 nodes.

100 Devices (BJTs, diodes, JFETs, and MOSFETs).

25 Independent voltage or current sources. Only 5 independent sources can be time dependent for transient analysis.

200 Total elements, including devices and independent sources.

10 Output variables. An output variable is either a node to node voltage or a current through an independent voltage source. Output variables may be printed in tabular form, plotted as line printer plots, or both. Only 5 output variables can be used in the AC small signal analysis.

20 Sets of model parameters for devices.

TYPES OF ANALYSES

DC Analysis

The DC analysis portion of SPICE determines the DC operating point of the circuit with inductors shorted and capacitors opened. A DC analysis is performed prior to a transient analysis to determine the transient initial conditions, and prior to an AC small signal analysis to determine the linearized, small signal models for nonlinear devices. If requested, the DC small signal value of a transfer function (ratio of output variable to input source), input impedance, and output impedance will also be computed as a part of the small signal operating point. The DC analysis can also be used to generate DC transfer curves. A specified independent voltage or current source is stepped over a user specified range and the DC output variables are stored for each sequential source value. The DC analysis options are specified on the .DC control statement (page 16).

NOTE: The user may suppress the DC operating point printout by using the .NP 1 control statement, described on page 14.

AC Small Signal Analysis

The AC small signal portion of SPICE computes the AC output variables as a function of frequency. The program first computes the DC operating point of the circuit and determines linearized, small signal models for all the nonlinear devices in the circuit. The resultant linear circuit is then analyzed over a user specified range of frequencies. The desired output of an AC small signal analysis is usually a transfer function (voltage gain, transimpedance, etc.). If the circuit has only one AC input, it is convenient to set that input to unity and zero phase, so that output variables have the same value as the transfer function of the output variable with respect to the input.

The generation of white noise by resistors and semiconductor devices can also be simulated with the AC small signal portion of SPICE. Equivalent noise source values are determined automatically from the small signal operating point of the circuit, and the contribution of each noise source is added at a given summing point. The total output noise level and the equivalent input noise level are determined at each frequency point. The output and input noise levels are normalized with respect to the square root of the noise bandwidth and have the units volts/$\sqrt{hz}$ or amps/$\sqrt{hz}$. The output noise and equivalent input noise can be printed or plotted in the same fashion as other output variables.

The frequency range and the noise analysis options are specified on the .AC control statement (page 17).

Transient Analysis

The transient analysis portion of SPICE computes the transient output variables as a function of time over a user specified time interval. The initial conditions are automatically determined by a DC analysis. All sources which are not time dependent (for example, power supplies) are set to their DC value. For large signal sinusoidal simulations, a Fourier analysis of the output waveform can be specified to obtain the frequency domain Fourier coefficients. The transient time interval and the Fourier analysis options are specified on the .TRAN control statement (page 17).

Analysis at Different Temperatures

All input data for SPICE is assumed to have been measured at 27° C (300° K). The simulation also assumes a nominal temperature of 27° C. The circuit can be simulated at up to 5 different temperatures by using a .TEMP control statement (page 15).

At present, the only values in SPICE which depend on temperature are the junction exponential terms,

EXP (Q * V / (N * K * TEMP))

and the junction saturation currents. The temperature dependence of the saturation current in the junction diode and the BJT models is determined by

IS (TEMP) = IO * (TEMP**(3/N)) * EXP (-Q * EG / (K * TEMP))

where IO, Q, and K are constants; N is the emission coefficient; and EG is the energy gap, which is a model parameter. For Shottky barrier diodes, the temperature dependence of the saturation current is determined by

IS (TEMP) = IC * (TEMP**(2/N)) * EXP (-Q * EG / (K * TEMP))

CONVERGENCE

Both DC and transient solutions are obtained by an iterative process which is terminated when the node voltages converge to within a tolerance of 0.1 percent or 50 microvolts. Although the particular algorithm used in SPICE has been found to be very reliable, in some cases it will fail to converge to a solution. When this happens, the program will print out the last node voltages and terminate the job. The node voltages that are printed are not necessarily correct or even close to the correct solution.

Failure to converge in the DC analysis is usually caused by an error in specifying circuit connections, element values, or model parameter values. Bistable circuits probably will not converge in the DC analysis. Failure to converge in the transient analysis can also be caused by a time step which is too large. SPICE presently does not have an automatic time step control, and significant error and/or nonconvergence can result if the time step is large compared with the circuit time constants.

INPUT FORMAT

The input format for SPICE is of the free format type. Fields in a statement are separated by one or more blanks, a comma (,), or an equal sign (=). Spaces preceding or following a comma or an equal sign are ignored. A statement may be continued onto the following line by placing a plus sign (+) before the first field on the continuation line.

A name field must begin with a letter (A through Z) and cannot contain commas or blanks. Only the first 5 characters of the name are used.

A number field may be an integer field (for example, 12,-44), a floating point field (for example, 3.14159), either an integer or a floating point number followed by an integer exponent (for example, 1E-14, 2.65E3), or either an integer or a floating point number followed by one of the following scale factors:

G	1.0E9
MEG	1.0E6
K	1.0E3
M	1.0E-3
U	1.0E-6
N	1.0E-9
P	1.0E-12

Letters immediately following a number that are not scale factors are ignored, and letters immediately following a scale factor are ignored. Hence, 10, 10V, 10VOLTS, and 10HZ all represent the same number, and M, MA, MSEC, and MMHOS all represent the same scale factor. Note that 1000, 1000.0, 1000HZ, 1E3, 1.0E3, 1KHZ, and 1K all represent the same number.

CIRCUIT DESCRIPTION

The circuit to be analyzed is described to SPICE by a set of circuit statements, which define the circuit topology and element values, and a set of control statements, which define the model parameters and the run controls. The first statement in the input file must be a title statement, and the last statement must be an .END statement. The order of the remaining statements is arbitrary.

Node numbers must be integers. The datum node must be numbered 0 (zero). Nodes need not be numbered sequentially. The circuit cannot contain a loop of voltage sources and/or inductors and cannot contain a cutset of current sources and/or capacitors. Each node in the circuit, including the datum node, must have at least two connections.

CIRCUIT STATEMENTS

Resistors, Capacitors, Inductors

General form:

```
RXXXX N1 N2 VALUE
CXXXX N1 N2 VALUE
LXXXX N1 N2 VALUE
```

Examples:

```
R13    12 17 1K
CGOOD 13 0 10P
LLINK  42 69 1U
```

N1 and N2 are the two element nodes. The order of the nodes for these elements is unimportant. Value is the resistance (ohms), the capacitance (farads), and the inductance (henries), respectively. This value cannot be negative or zero.

Voltage Controlled Current Sources

General form:

```
IXXXX V N+ N- NC+ NC- VALUE DELAY
```

Examples:

```
ISORS V 13 12 14 12 1.0M
IGM V 1 20 4 20 -2.0M 3.0NS
```

The letter V must be in the field following the element name. N+ and N- are the positive and negative nodes, respectively. Current flows from the positive node, through the source, to the negative node. NC+ and NC- are the positive and negative controlling nodes, respectively. Value is the transconductance (mhos).

In the AC analysis, the transconductance can be modified by an optional delay (linear phase) operator. The delay (seconds) is appended after the value. If a delay, TO, is included, the complex, frequency dependent value of transconductance is determined by

GM = VALUE * EXP(-J * 6.28318 * FREQ * TO)

The delay is ignored in the DC and transient analyses.

Independent Sources

General form:

```
VXXXX N+ N- DC DCVAL AC ACVAL PHASE
IXXXX N+ N- DC DCVAL AC ACVAL PHASE
```

Examples:

```
VCC 10 0 DC 6
IZENR 13 15 DC 600MA
VIN 13 2 DC 0.001 AC 1
IIN 21 23 AC 0.333 45.0
VMEAS 12 9
```

N+ is the positive node, and N- is the negative node. Note that voltage sources need not be grounded. Current flows from the positive node, through the source, to the negative node.

DCVAL is the DC value of the source. The source is set to this value for DC analysis and, if no time dependence is attached, in the transient analysis. If the DC source value is zero, the letters DC and the DC value can be omitted.

ACVAL is the AC value, and PHASE is the AC phase. The source is set to this value in the AC analysis. The arbitrary phase factor can be omitted. If the source is not an AC small signal input, the letters AC and the AC values are omitted.

A source may be given a time dependence for the transient analysis by appending one of the three predefined functions: PULSE, EXPONENTIAL, and SINUSOIDAL. If parameters other than source values are omitted or set to zero, the default values shown will be assumed. TSTEP is the printing increment (time step), and TSTOP is the final time (page 17).

PULSE

PULSE V1 V2 TD TR TF PW PER

Example: VIN 3 0 PULSE -1 1 2NS 2NS 2NS 50MS 100NS

Parameters and default values:

V1	Initial value	--
V2	Pulsed value	--
TD	Delay time	TSTEP
TR	Rise time	TSTEP
TF	Fall time	TSTEP
PW	Width	TSTOP
PER	Period	TSTOP

A single pulse is described by the following piecewise linear table:

Time	Value
0	V1
TD	V1
TD+TR	V2
TD+TR+PW	V2
TD+TR+PW+TF	V1
TSTOP	V1

EXPONENTIAL

EXP V1 V2 TD1 TAU1 TD2 TAU2

Example: VIN 3 0 EXP -4 -1 2NS 30NS 60NS 40NS

Parameters and default values:

V1	Initial value	--
V2	Pulsed value	--
TD1	Rise delay time	TSTEP
TAU1	Rise time constant	TSTEP
TD2	Fall delay time	TSTEP
TAU2	Fall time constant	TSTEP

Time	Value
0 to TD1	V1
TD1 to TD2	V1+(V2-V1) (1-EXP(-(T-TD1)/TAU1))
TD2 to TSTOP	V1+(V2-V1) (1-EXP(-(T-TD1)/TAU1)) +(V1-V2) (1-EXP(-(T-TD2)/TAU2))

SINUSOIDAL SIN VO VA FREQ TD THETA

Example: VIN 3 0 SIN 0 1 100MEG 1NS 1E10

Parameters and default values:

VO	Offset	--
VA	Amplitude	--
FREQ	Frequency (in hz)	1/TSTOP
TD	Delay	TSTEP
THETA	Damping factor	0

Time	Value
0 to TD	VO
TD to TSTOP	VO+VA*EXP(-(T-TD)*THETA)*SINE(6.28318*FREQ*T)

Sources may be given any combination of values (DC, AC, or transient), and these values may be specified in any order as long as they follow the proper key word.

Examples:
```
VIN 13 12 SIN 0 1 10MEG DC 0.1 AC 1 45
IZ 19 0 DC 0 PULSE 0 1 AC 0.5
VEQ 12 0 DC 0.5 EXP 0.5 0.9 10NS 40NS 70NS 40NS AC 1
```

Bipolar Junction Transistors

General form: QXXXX NC NB NE MNAME AREA

Example: QAMP33 7 9 1 MOD1 2.0

NC is the collector node, NB is the base node, NE is the emitter node, MNAME is the model name (page 9), and AREA is the area factor. The area factor is equivalent to the number of parallel devices. An area factor of 2.0 implies that two transistors of the same model are connected in parallel. If the area is omitted, an area factor of 1.0 is assumed.

Junction Diodes

General form: DXXXX N+ N- MNAME AREA

Example: DBRIG 8 10 DIODE

N+ is the positive node, N- is the negative node, MNAME is the model name, and AREA is the area factor (see BJTs, above).

Junction Field Effect Transistors

General form: JXXXX ND NG NS MNAME AREA

Example: J1 7 2 3 JM1

ND is the drain node, NG is the gate node, NS is the source node, MNAME is the model name, and AREA is the area factor (see BJTs, above).

MOSFETs

General form: MXXXX ND NG NS NB MNAME AREA

Example: M31G 2 3 4 7 MLONG

ND is the drain node, NG is the gate node, NS is the source node, NB is the bulk (substrate) node, MNAME is the model name, and AREA is the area factor (see BJTs, above).

.MODEL Statement

General form: .MODEL MNAME TYPE PNAME1=PVAL1 PNAME2=VAL2 ...

Example: .MODEL MOD1 NPN BF=50 IS=1E-13 VA=50

The .MODEL statement specifies a set of model parameters that will be used by one or more devices. MNAME is the model name, and TYPE is one of the following 10 types:

NPN	NPN Ebers-Moll BJT model
PNP	PNP Ebers-Moll BJT model
NGP	NPN Gummel-Poon BJT model
PGP	PNP Gummel-Poon BJT model
D	Junction diode model
SBD	Shottky barrier diode model
NJF	N channel JFET model
PJF	P channel JFET model
NMO	N channel MOSFET model
PMO	P channel MOSFET model

Parameter values are defined by appending the parameter name, as given below for each model type, followed by an equal sign and the parameter value. Model parameters that are not given a value are assigned the default value given below for each model type.

Model parameter values can also be specified as a string of numbers in the order given below for each model type. The following model specification is equivalent to the previous model statement example:

Example: .MODEL MOD1 NPN 50,,,,,,,,,,1E-13,,,50

Diode Models (both junction and SBD)

The only difference between the junction diode model and the Shottky barrier diode model is the temperature dependence of saturation current (see page 3) The DC characteristics of the diode are determined by the parameters IS and N. An ohmic resistance, RS, is included. Charge storage effects are modeled by a transit time, TT, and a nonlinear depletion layer capacitance which varies as the -1/2 power of junction voltage and is defined by the parameters CJO and PHI. The energy gap, EG, affects only the temperature dependence of the saturation current (see page 3).

	Name	Parameter	Default	Typical
1	RS	Ohmic resistance	0	10
2	TT	Transit time	0	0.1NS
3	CJO	Zero bias capacitance	0	2PF
4	IS	Saturation current	1.0E-14	1.0E-14
5	N	Emission coefficient	1	1
6	PHI	Junction potential	1	0.6
7	EG	Energy gap	1.11 SI 0.69 SBD	1.11 for SI 0.69 for SBD 0.67 for GE

Ebers-Moll BJT Models (both NPN and PNP)

The Ebers-Moll BJT model uses the DC Ebers-Moll model as a basis. The DC characteristics of the device are determined by the parameters BF and BR, the forward and reverse current gains, VA, which determines the output conductance, and the saturation current, IS. Three ohmic resistances, RB, RC, and RE, have been included. Base charge storage is modeled by forward and reverse transit times, TF and TR, and nonlinear depletion layer capacitances which vary as the -1/2 power of junction voltage and are defined by the parameters CJE, PE, CJC, and PC. A constant collector-substrate capacitance, CCS, is also included. The energy gap, EG, affects only the temperature dependence of the saturation current (see page 3).

	Name	Parameter	Default	Typical
1	BF	Forward beta	100	100
2	BR	Reverse beta	1	0.1
3	RB	Base ohmic resistance	0	100
4	RC	Collector ohmic resistance	0	10
5	RE	Emitter ohmic resistance	0	0.1
6	CCS	Collector-substrate capacitance	0	2PF
7	TF	Forward transit time	0	0.1NS
8	TR	Reverse transit time	0	10NS
9	CJE	Zero bias B-E capacitance	0	2PF
10	CJC	Zero bias B-C capacitance	0	1PF
11	IS	Saturation current	1.0E-14	1.0E-14
12	PE	B-E junction potential	1	0.7
13	PC	B-C junction potential	1	0.5
14	VA	Early voltage	INFINITE	50
15	EG	Energy gap	1.11	1.11 for SI 0.67 for GE

Gummel-Poon BJT Models (both NPN and PNP)

The integral charge model of Gummel and Poon is a more complicated and more complete BJT model for problems which require accurate BJT models. The DC model is defined by the parameters BFM, C2, IK, and NE, which determine the reverse current gain characteristics, BRM, C4, IKR, and NC, which determine the reverse current gain characteristics, VA and VB, which determine the output conductance for forward and reverse regions, and the saturation current, IS. Three ohmic resistances, RB, RC, and RE, are included. Base charge storage is modeled by forward and reverse transit times, TF and TR, and nonlinear depletion layer capacitances which are determined by CJE, PE, and ME for the B-C junction, and CJC, PC, and MC for the B-C junction. A constant collector-substrate capacitance, CCS, is also included. The energy gap, EG, is included as in the simpler BJT model.

	Name	Parameter	Default	Typical
1	BFM	Sat current/ideal B-E sat current	100	100
2	BRM	Sat current/ideal B-C sat current	1	0.1
3	RB	Base ohmic resistance	0	100
4	RC	Collector ohmic resistance	0	10
5	RE	Emitter ohmic resistance	0	0.1
6	CCS	Collector-substrate capacitance	0	2PF
7	TF	Forward transit time	0	0.1NS
8	TR	Reverse transit time	0	10NS
9	CJE	Zero bias B-E capacitance	0	2PF
10	CJC	Zero bias B-C capacitance	0	1PF
11	IS	Saturation current	1.0E-14	1.0E-14
12	VA	Forward early voltage	INFINITE	50
13	VB	Reverse early voltage	INFINITE	50
14	C2	Nonideal B-E sat current/sat current	0	10.0
15	IK	Forward knee current	INFINITE	10MA
16	NE	B-E emission coefficient	2.0	1.5
17	C4	Nonideal B-C sat current/sat current	0	1.0
18	IKR	Reverse knee current	INFINITE	100MA
19	NC	B-C emission coefficient	2.0	1.5
20	PE	B-E junction potential	1.0	0.7
21	ME	B-E grading coefficient	0.5	0.33
22	PC	B-C junction potential	1.0	0.5
23	MC	B-C grading coefficient	0.5	0.33
24	EG	Energy gap	1.11	1.11 for SI 0.67 for GE

JFET Models (both N and P channel)

The JFET model is derived from the FET model of Hodges and Schichman. The DC characteristics are defined by the parameters VTO and BETA, which determine the variation of drain current with gate voltage, LAMBDA, which determines the output conductance, and IS, the saturation current of the two gate junctions. Two ohmic resistances, RD and RS, are included. Charge storage is modeled by nonlinear depletion layer capacitances for both gate junctions which vary as the -1/2 power of junction voltage and are defined by the parameters CGS, CGD, and PB.

	Name	Parameter	Default	Typical
1	VTO	Threshold voltage	-2.0	-2.0
2	BETA	Transconductance parameter	1.0E-4	1.0E-4
3	LAMBDA	Channel length modulation parameter	0	0.01
4	RD	Drain ohmic resistance	0	100
5	RS	Source ohmic resistance	0	100
6	CGS	Zero bias gate-source capacitance	0	2PF
7	CGD	Zero bias gate-drain capacitance	0	2PF
8	PB	Gate junction potential	1	0.6
9	IS	Gate junction saturation current	1.0E-14	1.0E-14

MOSFET Models (both N and P channels)

The MOSFET model is also derived from the FET model of Hodges and Schichman. The DC characteristics of the MOSFET are defined by the parameters VTO, BETA, and LAMBDA, which are identical to the parameters for the JFET, PHI, and GAMMA, which determine the variation of threshold voltage with substrate voltage, and IS, the saturation current of the two substrate junctions. Charge storage is modeled by three linear capacitors, CGS, CGD, and CGB, and nonlinear depletion layer capacitances for both substrate junctions which vary as the -1/2 power of junction voltage and are determined by the parameters CBD, CBS, and PB.

	Name	Parameter	Default	Typical
1	VTO	Threshold voltage	2.0	2.0
2	PHI	Surface potential	0.5	0.5
3	BETA	Transconductance parameter	1.0E-4	1.0E-4
4	GAMMA	Bulk threshold parameter	0	0.5
5	LAMBDA	Channel length modulation parameter	0	0.01
6	RD	Drain ohmic resistance	0	100
7	RS	Source ohmic resistance	0	100
8	CGS	Gate-source capacitance	0	1PF
9	CGD	Gate-drain capacitance	0	1PF
10	CGB	Gate-bulk capacitance	0	1PF
11	CBD	Zero bias bulk-drain capacitance	0	1PF
12	CBS	Zero bias bulk-source capacitance	0	1PF
13	PB	Bulk junction potential	1	0.6
14	IS	Bulk junction saturation current	1.0E-14	1.0E-14

CONTROL STATEMENTS

Title Statement

Example: OP AMP CIRCUIT JOE J STUDENT EECS 241

This statement must be the first line in the input file. Its contents are printed verbatim as the heading for each section of output.

A period (.) as the first character of the title card suppresses the listing of the input file.

.END Statement

General form: .END

This statement must always be the last statement in the input file. Note that the period is an integral part of the name. If the .END statement is omitted, the job will not run successfully.

Comment Statement

General form: * ANY COMMENTS

Example: * RF=1K GAIN SHOULD BE 100

This statement is printed out in the input listing, but is otherwise ignored.

No Print Statement

General form: .NP

This statement suppresses the summary of input data that is normally printed after reading the input file. It does not suppress the listing of the input file or any error messages that may occur.

If the digit 1 is appended to the .NP statement, as

.NP 1

SPICE also suppresses the DC operating point printout which precedes the transient, noise sensitivity, Fourier, and AC small signal analysis.

.TEMP Statement

General form: .TEMP TE1 TE2 ...

Example: .TEMP -55.0 25.0 125.0

This statement specifies the temperatures at which the circuit is to be simulated. TE1, TE2,... are the different temperatures, in degrees C. A maximum of five temperatures is allowed. Temperatures less than -223.0° C are ignored.

.OUTPUT Statement

General form:
```
.OUTPUT VXXXX  N+  N-
.OUTPUT IXXXX VYYYY
.OUTPUT NOISE
```

Examples:
```
.OUTPUT VMIXE 13 27
.OUTPUT IBASE BV17
```

This statement defines an output variable. For voltage outputs, the name must begin with a V, and N+ and N- are the positive and negative nodes of the output voltage. For current outputs, the output name must begin with an I, and VYYYY is the name of the independent voltage source that the current is flowing in. Positive current flows from the positive node, through the source, to the negative node. The output variable name noise is reserved for the noise analysis, and the output noise and equivalent input noise can be printed and plotted in the same fashion as other output variables.

Outputs can be printed in tabular form or plotted as line printer plots. There are eight different options which can be printed and/or plotted:

DC	DC transfer curve output
TR	Transient analysis output
RE	AC analysis output, real part
IM	AC analysis output, imaginary part
MA	AC analysis output, magnitude
PH	AC analysis output, phase
OU	Noise analysis output, total output noise voltage
IN	Noise analysis output, equivalent input noise

An output can be printed or plotted by appending the letters PRINT or PLOT, followed by any combination of the eight output options, to the .OUTPUT statement.

Examples:

```
.OUTPUT V13 13 0 PRINT MA DC TR
.OUTPUT IIN VIN PRINT PH RE DC
.OUTPUT VOUT 17 2 PLOT MA TR PRINT DC
.OUTPUT II3 V13 PLOT PH DC
.OUTPUT VTHRE 3 0 PRINT DC PLOT TRAN
.OUTPUT NOISE PRINT IN PLOT OU
```

The program will automatically determine the minimum and maximum values of the output variable and scale the plot to fit these limits. The automatic scaling feature can be overridden by specifying plot limits after the output option. The plot limits apply only to the option that they follow.

Example:

```
.OUTPUT V12 12 0 PLOT MA PH -20 30 TR 0 5
```

In this example, the program will determine limits for the magnitude plot, but will plot the phase between -20 degrees and 30 degrees, and will plot the transient response between 0 volts and 5 volts.

.DC Statement

General form:

```
.DC OP OUTPUT INPUT TC ELNAME VSTART VSTOP VINCR
```

Examples:

```
.DC OP
.DC TC VIN 0 5 0.5
.DC OP VOUT VIN TC VIN 0 5 0.5
```

For the small signal transfer function, OUTPUT is the output variable, and INPUT is the input source. The program will compute the DC small signal value of the transfer function (OUTPUT/INPUT), input impedance, and output impedance. If the transfer function value is not desired, the user omits the OUTPUT and INPUT specifications. If the DC operating point is not desired, the user omits the letters OP. However, a DC operating point will always be computed prior to an AC small signal analysis or a transient analysis.

For transfer curves, ELNAME is the name of the variable source, VSTART is the starting source value, VSTOP is the final source value, and VINCR is the increment. The total number of points to be computed cannot exceed 101. If a transfer curve is not desired, the user omits the letters TC and the transfer curve parameters.

.AC Statement

General form:

```
.AC DEC ND FSTART FSTOP NOISE OUTPUT INPUT NUMS
.AC OCT NO FSTART FSTOP NOISE OUTPUT INPUT NUMS
.AC LIN NP FSTART FSTOP NOISE OUTPUT INPUT NUMS
```

Examples:

```
.AC DEC 10 1 10KHZ
.AC DEC 20 1 100KHZ NOISE VOUT VIN 10
.AC DEC 10 1 100MEG NOISE VOUT V21
```

DEC stands for decade variation, and ND is the number of points per decade. OCT stands for octave variation, and NO is the number of points per octave. LIN stands for linear, and NP is the number of points. FSTART is the starting frequency, and FSTOP is the final frequency. The total number of frequency points to be computed cannot exceed 101. For noise analysis, OUTPUT is the name of a voltage output variable. This output, which must be a voltage, will be used as the summing point. INPUT is the name of an independent voltage or current source. The total output noise is divided by the transfer function (output/input) to obtain the equivalent input noise level. NUMS is the summary interval. At every NUMS frequency point, the individual contributions of each element are printed. If NUMS is omitted or set to zero, no summary printout will occur. For reasons of reducing printout, NUMS should be as large as possible. If the noise analysis is not desired, omit the letters NOISE and the noise analysis specifications.

.TRAN Statement

General form: .TRAN TSTEP TSTOP TSTART FOUR OUTPUT FREQ

Examples:

```
.TRAN 1NS 100NS
.TRAN 1NS 1000NS 500NS
.TRAN 1NS 100NS FOUR VOUT 100MEG
```

TSTEP is the printing increment between timepoints; TSTOP is the final timepoint; and TSTART is the initial timepoint. If TSTART is omitted, it is assumed to be zero. The transient analysis always begins at time zero. In the interval (zero, TSTART), the circuit is analyzed (to reach a steady state), but no outputs are stored. In the interval (TSTART, TSTOP), the circuit is analyzed and outputs are stored. The number of timepoints in the interval (TSTART, TSTOP) cannot exceed 101.

For Fourier analysis, OUTPUT is the output variable, and FREQ is the fundamental frequency. The Fourier analysis is performed over the interval (TSTOP-PERIOD,TSTOP), where TSTOP is the final time specified, and PERIOD is one period of the fundamental frequency. The DC component and the first nine components are determined. For maximum accuracy, the number of periods in the interval (TSTART,TSTOP) should be as small as possible (but never less than one). This ensures that the number of timepoints in one fundamental is as large as possible. If the Fourier analysis is not desired, the user omits the letters FOUR and the Fourier specifications.

For some problems, to avoid numerical instability in the integration algorithm, it may be necessary to specify an internal time step which is smaller than the printing increment (TSTEP). Examples of this type of problem are astable multivibrators, sweep circuits, and other highly nonlinear circuits which have widely separated time constants. SPICE allows the user to segment the time interval into from one to five subintervals and specify a different time step for each subinterval. The interval time steps and subinterval endpoints are specified after the starting time (TSTART) and before the Fourier analysis options.

General form:

```
.TRAN TSTEP TSTOP TSTART D1 E1 D2 E2 ... D5 E5 FOUR OUTPUT FRE
```

Example: `.TRAN 1NS 100NS 0 0.1NS 10NS 0.5NS 100NS`

D1 is the first internal time step, and E1 is the endpoint of the first subinterval; D2 is the second internal time step, and E2 is the endpoint of the second subinterval; and so on. In this example, the program will use an internal time step of 0.1 ns for the interval (0,10NS) and an internal time step of 0.5 ns for the interval (10NS,100NS). Output is still stored every 1 ns. The total number of timepoints to be computed cannot exceed 1001.

Example: `.TRAN 1US 100US 0 0.1US 100US`

In this example, the program will use an internal time step of 0.1US over the entire transient interval, but will store output only at 1US intervals. Hence, the program stores and outputs every tenth timepoint.

SAMPLE DATA FILES

The following file determines the DC operating point and small signal transfer function of a simple differential pair.

```
SIMPLE DIFFERENTIAL PAIR
VCC 7 0 DC 12
VEE 8 0 DC -12
VIN 1 0
RS1 1 2 1K
RS2 6 0 1K
Q1 3 2 4 MOD1
Q2 5 6 4 MOD1
RC1 7 3 10K
RC2 7 5 10K
RE 4 8 10K
.MODEL MOD1 NPN BR=50 VA=50 IS=1.0E-12 RB=100
.OUT VOUT 5 0
.DC OP VOUT VIN
.END
```

The following file determines the DC transfer curve and the transient pulse response of a simple RTL inverter. The input is a pulse from 0 to 5 volts with delay, rise, and fall times of 2 ns and a pulse width of 30 ns. The transient interval is 0 to 100 ns in 1 ns steps.

```
SIMPLE RTL INVERTER
VCC 4 0 DC 5
VIN 1 0 PULSE 0 5 2NS 2NS 2NS 30NS
RB 1 2 10K
Q1 3 2 0 Q1
RC 4 3 1K
.OUTPUT VC 3 0 PRINT DC PLOT TR 0 5
.MODEL Q1 NPN BF=20 RB=100 TF=0.1NS CJC=2PF
.DC TC VIN 0 5 0.1
.TRAN 1NS 100NS
.END
```

The following file determines the AC small signal response of a one transistor amplifier over the frequency range of 1HZ to 100MEGHZ.

```
ONE TRANSISTOR AMPLIFIER
VCC 5 0 DC 12
VEE 6 0 DC -12
VIN 1 0 AC 1
RS 1 2 1K
Q1 3 2 4 X33
RC 5 3 500
RE 4 6 1K
CBYPASS 4 0 1UFD
.OUT V3 3 0 PLOT MA PH
.AC DEC 10 1HZ 100MEGHZ
.MODEL X33 NPN BF=30 RB=50 VA=20
.END
```

SAMPLE PROBLEM

The example below illustrates the simple procedure for creating an input file for SPICE and executing that file. Everything typed by the user is underlined, and the symbol for a user-typed Carriage Return is ↩.

The problem analyzes the circuit in the sample data file above, determining the AC small signal response of a one transistor amplifier over the frequency range of 1HZ to 100MEGHZ.

```
-EDITOR↩
*APPEND↩
ONE TRANSISTOR AMPLIFIER↩
VCC 5 0 DC 12↩
VEE 6 0 DC -12↩
VIN 1 0 AC 1↩
RS 1 2 1K↩
Q1 3 2 4 X33↩
RC 5 3 500↩
RE 4 6 1K↩
CBYPASS 4 0 1UPD↩
.OUT V3 3 0 PLOT MA PH↩
.AC DEC 10 1HZ 100MEGHZ↩
.MODEL X33 NPN BF=30 RB=50 VA=20↩
.END↩
```

The user creates the SPICE input file in EDITOR using the APPEND command.

The .END statement must be the final line in the file.

```
*WRITE AMP↩
 NEW FILE↩
230 CHRS
*QUIT↩
```

Control D ends the APPEND command. The user writes the contents of EDITOR onto the file AMP.

```
-R SPICE↩
```

He calls SPICE and specifies AMP as the input file.

```
INPUT FROM: AMP↩

OUPUT TO: T↩
```

He may specify T for terminal output or a file name for file output.

```
ONE TRANSISTOR AMPLIFIER
VCC 5 0 DC 12
VEE 6 0 DC -12
VIN 1 0 AC 1
RS 1 2 1K
Q1 3 2 4 X33
RC 5 3 500
RE 4 6 1K
CBYPASS 4 0 1UPD
.OUT V3 3 0 PLOT MA PH
.AC DEC 10 1HZ 100MEGHZ
.MODEL X33 NPN BF=30 RB=50 VA=20
.END
```

The circuit summary and analysis are then printed at the terminal or on the specified file.

C I R C U I T   S U M M A R Y

*****RESISTORS, CAPACITORS, AND INDUCTORS

NAME	NODES		VALUE
RS	1	2	1.00E+03
RC	5	3	5.00E+02
RE	4	6	1.00E+03
CBYPA	4	0	1.00E-06

*****BIPOLAR JUNCTION TRANSISTORS

NAME	C	B	E	MODEL	AREA
Q1	3	2	4	X33	1.000

*****INDEPENDENT SOURCES

NAME	+	-	DC VALUE	AC VALUE	AC PHASE
VCC	5	0	1.20E+01	0.00E-01	0.00E-01
VEE	6	0	-1.20E+01	0.00E-01	0.00E-01
VIN	1	0	0.00E-01	1.00E+00	0.00E-01

*****NODE TABLE

NODE	ELEMENTS CONNECTED			
0	CBYPA	VCC	VEE	VIN
1	RS	VIN		
2	RS	Q1		
3	RC	Q1		
4	RE	CBYPA	Q1	
5	RC	VCC		
6	RE	VEE		

*****EBERS-MOLL MODEL PARAMETERS

NAME	X33
TYPE	N
BF	3.00E+01
BR	1.00E+00
RB	5.00E+01
RC	0.00E-01
RE	0.00E-01
CCS	0.00E-01
TF	0.00E-01
TR	0.00E-01
CJE	0.00E-01
CJC	0.00E-01
IS	1.00E-14
PE	1.00E+00
PC	1.00E+00
VA	2.00E+01
EG	1.11E+00

```
*****OUTPUT REQUESTS

NAME  ELEMENT      +   -

V3                 3   0

*****FREQUENCY VARIATION

FIRST FREQ      LAST FREQ      VARIATION      POINTS
 1.00E+00        1.00E+08         DEC           10

*****CIRCUIT TEMPERATURES (DEG C)

   27.00

MATRIX STATISTICS

NODES   NSTOP   NUTBR   NUTAR   IFILL  NSTERM  NTTERM   IOPS    PERSPA

   7       4       4       4       0       3      21       6    57.143

*****SMALL SIGNAL BIAS SOLUTION     TEMPERATURE     27.000  DEG C

 NODE    VOLTAGE     NODE    VOLTAGE     NODE    VOLTAGE     NODE    VOLTAGE

(  1)     0.0000   (  2)    -0.2662   (  3)     6.6271   (  4)    -0.9880
(  5)    12.0000   (  6)   -12.0000

*****VOLTAGE SOURCE CURRENTS

NAME  CURRENT

VCC    -1.075E-02  AMPS
VEE     1.101E-02  AMPS
VIN    -2.662E-04  AMPS

TOTAL POWER DISSIPATION   2.61E-01  WATTS

*****TRANSISTOR OPERATING POINTS

Q1    : IB= 2.68E-04  IC= 1.08E-02  VBE=  0.722  VBC= -6.893
        VCE=  7.615  BETADC=   40.4  BETAAC=   40.3  GM= 4.18E-01
        RPI=9.65E+01  RO=2.49E+03  CPI=0.00E-01  CMU=0.00E-01
        FT=6.65E+18
```

*****AC ANALYSIS TEMPERATURE 27.000 DEG C

FREQUENCY MAGNITUDE OF V3




1.995E+04
2.512E+04
3.162E+04
3.981E+04
5.012E+04
6.310E+04
7.943E+04
1.000E+05
1.259E+05
1.585E+05
1.995E+05
2.512E+05
3.162E+05
3.981E+05
5.012E+05
6.310E+05
7.943E+05
1.000E+06
1.259E+06
1.585E+06
1.995E+06
2.512E+06
3.162E+06
3.981E+06
5.012E+06
6.310E+06
7.943E+06
1.000E+07
1.259E+07
1.585E+07
1.995E+07
2.512E+07
3.162E+07
3.981E+07
5.012E+07
6.310E+07
7.943E+07
1.000E+08

```
*****AC ANALYSIS     TEMPFRATURE    27.000 DEG C

 FREQUENCY                PHASE OF V3    DEGRFES

          -1.800E+02                                  1.800E+02
            +....+....+....+....+....+....+....+....+....+....+
 1.000E+00  +
 1.259E+00  +
 1.585E+00  +
 1.995E+00  +
 2.512E+00  +
 3.162E+00  +
 3.981E+00  +
 5.012E+00  +
 6.310E+00  +
 7.943E+00  +
 1.000E+01  +
 1.259E+01   +
 1.585E+01   +
 1.995E+01   +
 2.512E+01   +
 3.162E+01    +
 3.981E+01    +
 5.012E+01    +
 6.310E+01     +
 7.943E+01      +
 1.000E+02      +
 1.259E+02       +
 1.585E+02        +
 1.995E+02         +
 2.512E+02          +
 3.162E+02          +
 3.981E+02           +
 5.012E+02           +
 6.310E+02            +
 7.943E+02            +
 1.000F+03            +
 1.259E+03            +
 1.585E+03           +
 1.995E+03           +
 2.512E+03          +
 3.162E+03          +
 3.981E+03         +
 5.012E+03        +
 6.310E+03       +
 7.943E+03      +
 1.000E+04      +
 1.259E+04     +
 1.585E+04    +
 1.995F+04    +
 2.512E+04    +
 3.162E+04   +
```

```
  3.981E+04    +
  5.012E+04    +
  6.310E+04    +
  7.943E+04   +
  1.000E+05   +
  1.259E+05   +
  1.585E+05   +
  1.995E+05   +
  2.512E+05   +
  3.162E+05   +
  3.981E+05   +
  5.012E+05   +
  6.310E+05   +
  7.943E+05   +
  1.000E+06   +
  1.259E+06   +
  1.585E+06   +
  1.995E+06   +
  2.512E+06   +
  3.162E+06   +
  3.981E+06   +
  5.012E+06   +
  6.310E+06   +
  7.943E+06   +
  1.000E+07   +
  1.259E+07   +
  1.585E+07   +
  1.995E+07   +
  2.512E+07   +
  3.162E+07   +
  3.981E+07   +
  5.012E+07   +
  6.310E+07   +
  7.943E+07   +
  1.000E+08   +
              +....+....+....+....+....+....+....+....+....+....+
EXIT

-
```